=head1 NAME
X
perlop - Perl operators and precedence
=head1 DESCRIPTION
=head2 Operator Precedence and Associativity
X X X
Operator precedence and associativity work in Perl more or less like
they do in mathematics.
I means some operators are evaluated before
others. For example, in C<2 + 4 * 5>, the multiplication has higher
precedence so C<4 * 5> is evaluated first yielding C<2 + 20 ==
22> and not C<6 * 5 == 30>.
I defines what happens if a sequence of the
same operators is used one after another: whether the evaluator will
evaluate the left operations first or the right. For example, in C<8
- 4 - 2>, subtraction is left associative so Perl evaluates the
expression left to right. C<8 - 4> is evaluated first making the
expression C<4 - 2 == 2> and not C<8 - 2 == 6>.
Perl operators have the following associativity and precedence,
listed from highest precedence to lowest. Operators borrowed from
C keep the same precedence relationship with each other, even where
C's precedence is slightly screwy. (This makes learning Perl easier
for C folks.) With very few exceptions, these all operate on scalar
values only, not array values.
left terms and list operators (leftward)
left ->
nonassoc ++ --
right **
right ! ~ \ and unary + and -
left =~ !~
left * / % x
left + - .
left << >>
nonassoc named unary operators
nonassoc < > <= >= lt gt le ge
nonassoc == != <=> eq ne cmp ~~
left &
left | ^
left &&
left || //
nonassoc .. ...
right ?:
right = += -= *= etc.
left , =>
nonassoc list operators (rightward)
right not
left and
left or xor
In the following sections, these operators are covered in precedence order.
Many operators can be overloaded for objects. See L.
=head2 Terms and List Operators (Leftward)
X X X
A TERM has the highest precedence in Perl. They include variables,
quote and quote-like operators, any expression in parentheses,
and any function whose arguments are parenthesized. Actually, there
aren't really functions in this sense, just list operators and unary
operators behaving as functions because you put parentheses around
the arguments. These are all documented in L.
If any list operator (print(), etc.) or any unary operator (chdir(), etc.)
is followed by a left parenthesis as the next token, the operator and
arguments within parentheses are taken to be of highest precedence,
just like a normal function call.
In the absence of parentheses, the precedence of list operators such as
C, C, or C is either very high or very low depending on
whether you are looking at the left side or the right side of the operator.
For example, in
@ary = (1, 3, sort 4, 2);
print @ary; # prints 1324
the commas on the right of the sort are evaluated before the sort,
but the commas on the left are evaluated after. In other words,
list operators tend to gobble up all arguments that follow, and
then act like a simple TERM with regard to the preceding expression.
Be careful with parentheses:
# These evaluate exit before doing the print:
print($foo, exit); # Obviously not what you want.
print $foo, exit; # Nor is this.
# These do the print before evaluating exit:
(print $foo), exit; # This is what you want.
print($foo), exit; # Or this.
print ($foo), exit; # Or even this.
Also note that
print ($foo & 255) + 1, "\n";
probably doesn't do what you expect at first glance. The parentheses
enclose the argument list for C which is evaluated (printing
the result of C). Then one is added to the return value
of C (usually 1). The result is something like this:
1 + 1, "\n"; # Obviously not what you meant.
To do what you meant properly, you must write:
print(($foo & 255) + 1, "\n");
See L for more discussion of this.
Also parsed as terms are the C and C constructs, as
well as subroutine and method calls, and the anonymous
constructors C and C.
See also L toward the end of this section,
as well as L"I/O Operators">.
=head2 The Arrow Operator
X X X<< -> >>
"C<< -> >>" is an infix dereference operator, just as it is in C
and C++. If the right side is either a C, C, or a
C subscript, then the left side must be either a hard or
symbolic reference to an array, a hash, or a subroutine respectively.
(Or technically speaking, a location capable of holding a hard
reference, if it's an array or hash reference being used for
assignment.) See L and L.
Otherwise, the right side is a method name or a simple scalar
variable containing either the method name or a subroutine reference,
and the left side must be either an object (a blessed reference)
or a class name (that is, a package name). See L.
=head2 Auto-increment and Auto-decrement
X X X X X X
"++" and "--" work as in C. That is, if placed before a variable,
they increment or decrement the variable by one before returning the
value, and if placed after, increment or decrement after returning the
value.
$i = 0; $j = 0;
print $i++; # prints 0
print ++$j; # prints 1
Note that just as in C, Perl doesn't define B the variable is
incremented or decremented. You just know it will be done sometime
before or after the value is returned. This also means that modifying
a variable twice in the same statement will lead to undefined behaviour.
Avoid statements like:
$i = $i ++;
print ++ $i + $i ++;
Perl will not guarantee what the result of the above statements is.
The auto-increment operator has a little extra builtin magic to it. If
you increment a variable that is numeric, or that has ever been used in
a numeric context, you get a normal increment. If, however, the
variable has been used in only string contexts since it was set, and
has a value that is not the empty string and matches the pattern
C^[a-zA-Z]*[0-9]*\z/>, the increment is done as a string, preserving each
character within its range, with carry:
print ++($foo = '99'); # prints '100'
print ++($foo = 'a0'); # prints 'a1'
print ++($foo = 'Az'); # prints 'Ba'
print ++($foo = 'zz'); # prints 'aaa'
C is always treated as numeric, and in particular is changed
to C<0> before incrementing (so that a post-increment of an undef value
will return C<0> rather than C).
The auto-decrement operator is not magical.
=head2 Exponentiation
X X X
Binary "**" is the exponentiation operator. It binds even more
tightly than unary minus, so -2**4 is -(2**4), not (-2)**4. (This is
implemented using C's pow(3) function, which actually works on doubles
internally.)
=head2 Symbolic Unary Operators
X X
Unary "!" performs logical negation, i.e., "not". See also C for a lower
precedence version of this.
X
Unary "-" performs arithmetic negation if the operand is numeric,
including any string that looks like a number. If the operand is
an identifier, a string consisting of a minus sign concatenated
with the identifier is returned. Otherwise, if the string starts
with a plus or minus, a string starting with the opposite sign is
returned. One effect of these rules is that -bareword is equivalent
to the string "-bareword". If, however, the string begins with a
non-alphabetic character (excluding "+" or "-"), Perl will attempt to convert
the string to a numeric and the arithmetic negation is performed. If the
string cannot be cleanly converted to a numeric, Perl will give the warning
B.
X X
Unary "~" performs bitwise negation, i.e., 1's complement. For
example, C<0666 & ~027> is 0640. (See also L and
L.) Note that the width of the result is
platform-dependent: ~0 is 32 bits wide on a 32-bit platform, but 64
bits wide on a 64-bit platform, so if you are expecting a certain bit
width, remember to use the & operator to mask off the excess bits.
X X
Unary "+" has no effect whatsoever, even on strings. It is useful
syntactically for separating a function name from a parenthesized expression
that would otherwise be interpreted as the complete list of function
arguments. (See examples above under L.)
X
Unary "\" creates a reference to whatever follows it. See L
and L. Do not confuse this behavior with the behavior of
backslash within a string, although both forms do convey the notion
of protecting the next thing from interpolation.
X X X
=head2 Binding Operators
X X X<=~> X
Binary "=~" binds a scalar expression to a pattern match. Certain operations
search or modify the string $_ by default. This operator makes that kind
of operation work on some other string. The right argument is a search
pattern, substitution, or transliteration. The left argument is what is
supposed to be searched, substituted, or transliterated instead of the default
$_. When used in scalar context, the return value generally indicates the
success of the operation. The exceptions are substitution (s///)
and transliteration (y///) with the C (non-destructive) option,
which cause the Beturn value to be the result of the substitution.
Behavior in list context depends on the particular operator.
See L"Regexp Quote-Like Operators"> for details and L for
examples using these operators.
If the right argument is an expression rather than a search pattern,
substitution, or transliteration, it is interpreted as a search pattern at run
time. Note that this means that its contents will be interpolated twice, so
'\\' =~ q'\\';
is not ok, as the regex engine will end up trying to compile the
pattern C, which it will consider a syntax error.
Binary "!~" is just like "=~" except the return value is negated in
the logical sense.
Binary "!~" with a non-destructive substitution (s///r) or transliteration
(y///r) is a syntax error.
=head2 Multiplicative Operators
X
Binary "*" multiplies two numbers.
X
Binary "/" divides two numbers.
X> X
Binary "%" is the modulo operator, which computes the division
remainder of its first argument with respect to its second argument.
Given integer
operands C and C: If C is positive, then C is
C minus the largest multiple of C less than or equal to
C. If C is negative, then C is C minus the
smallest multiple of C that is not less than C (i.e. the
result will be less than or equal to zero). If the operands
C and C are floating point values and the absolute value of
C (that is C) is less than C, only
the integer portion of C and C will be used in the operation
(Note: here C means the maximum of the unsigned integer type).
If the absolute value of the right operand (C) is greater than
or equal to C, "%" computes the floating-point remainder
C in the equation C where C is a certain
integer that makes C have the same sign as the right operand
C (B as the left operand C like C function C)
and the absolute value less than that of C.
Note that when C is in scope, "%" gives you direct access
to the modulo operator as implemented by your C compiler. This
operator is not as well defined for negative operands, but it will
execute faster.
X X X X
Binary "x" is the repetition operator. In scalar context or if the left
operand is not enclosed in parentheses, it returns a string consisting
of the left operand repeated the number of times specified by the right
operand. In list context, if the left operand is enclosed in
parentheses or is a list formed by C, it repeats the list.
If the right operand is zero or negative, it returns an empty string
or an empty list, depending on the context.
X
print '-' x 80; # print row of dashes
print "\t" x ($tab/8), ' ' x ($tab%8); # tab over
@ones = (1) x 80; # a list of 80 1's
@ones = (5) x @ones; # set all elements to 5
=head2 Additive Operators
X
Binary "+" returns the sum of two numbers.
X
Binary "-" returns the difference of two numbers.
X
Binary "." concatenates two strings.
X X
X X X X
=head2 Shift Operators
X X X<<< << >>>
X<<< >> >>> X X X
X X X X
Binary "<.)
Binary ">>" returns the value of its left argument shifted right by
the number of bits specified by the right argument. Arguments should
be integers. (See also L.)
Note that both "<>" in Perl are implemented directly using
"<>" in C. If C (see L) is
in force then signed C integers are used, else unsigned C integers are
used. Either way, the implementation isn't going to generate results
larger than the size of the integer type Perl was built with (32 bits
or 64 bits).
The result of overflowing the range of the integers is undefined
because it is undefined also in C. In other words, using 32-bit
integers, C<< 1 << 32 >> is undefined. Shifting by a negative number
of bits is also undefined.
=head2 Named Unary Operators
X
The various named unary operators are treated as functions with one
argument, with optional parentheses.
If any list operator (print(), etc.) or any unary operator (chdir(), etc.)
is followed by a left parenthesis as the next token, the operator and
arguments within parentheses are taken to be of highest precedence,
just like a normal function call. For example,
because named unary operators are higher precedence than ||:
chdir $foo || die; # (chdir $foo) || die
chdir($foo) || die; # (chdir $foo) || die
chdir ($foo) || die; # (chdir $foo) || die
chdir +($foo) || die; # (chdir $foo) || die
but, because * is higher precedence than named operators:
chdir $foo * 20; # chdir ($foo * 20)
chdir($foo) * 20; # (chdir $foo) * 20
chdir ($foo) * 20; # (chdir $foo) * 20
chdir +($foo) * 20; # chdir ($foo * 20)
rand 10 * 20; # rand (10 * 20)
rand(10) * 20; # (rand 10) * 20
rand (10) * 20; # (rand 10) * 20
rand +(10) * 20; # rand (10 * 20)
Regarding precedence, the filetest operators, like C, C, etc. are
treated like named unary operators, but they don't follow this functional
parenthesis rule. That means, for example, that C is
equivalent to C.
X X X
See also L.
=head2 Relational Operators
X X
Binary ">
Binary ">" returns true if the left argument is numerically greater
than the right argument.
X<< > >>
Binary "<=" returns true if the left argument is numerically less than
or equal to the right argument.
X<< <= >>
Binary ">=" returns true if the left argument is numerically greater
than or equal to the right argument.
X<< >= >>
Binary "lt" returns true if the left argument is stringwise less than
the right argument.
X<< lt >>
Binary "gt" returns true if the left argument is stringwise greater
than the right argument.
X<< gt >>
Binary "le" returns true if the left argument is stringwise less than
or equal to the right argument.
X<< le >>
Binary "ge" returns true if the left argument is stringwise greater
than or equal to the right argument.
X<< ge >>
=head2 Equality Operators
X X X X
Binary "==" returns true if the left argument is numerically equal to
the right argument.
X<==>
Binary "!=" returns true if the left argument is numerically not equal
to the right argument.
X
Binary "<=>" returns -1, 0, or 1 depending on whether the left
argument is numerically less than, equal to, or greater than the right
argument. If your platform supports NaNs (not-a-numbers) as numeric
values, using them with "<=>" returns undef. NaN is not "",
"<=" or ">=" anything (even NaN), so those 5 return false. NaN != NaN
returns true, as does NaN != anything else. If your platform doesn't
support NaNs then NaN is just a string with numeric value 0.
X<< <=> >> X
perl -le '$a = "NaN"; print "No NaN support here" if $a == $a'
perl -le '$a = "NaN"; print "NaN support here" if $a != $a'
Binary "eq" returns true if the left argument is stringwise equal to
the right argument.
X
Binary "ne" returns true if the left argument is stringwise not equal
to the right argument.
X
Binary "cmp" returns -1, 0, or 1 depending on whether the left
argument is stringwise less than, equal to, or greater than the right
argument.
X
Binary "~~" does a smart match between its arguments. Smart matching
is described in L.
X
"lt", "le", "ge", "gt" and "cmp" use the collation (sort) order specified
by the current locale if C is in effect. See L.
=head2 Bitwise And
X X X
Binary "&" returns its operands ANDed together bit by bit.
(See also L and L.)
Note that "&" has lower priority than relational operators, so for example
the brackets are essential in a test like
print "Even\n" if ($x & 1) == 0;
=head2 Bitwise Or and Exclusive Or
X X X X
X X
Binary "|" returns its operands ORed together bit by bit.
(See also L and L.)
Binary "^" returns its operands XORed together bit by bit.
(See also L and L.)
Note that "|" and "^" have lower priority than relational operators, so
for example the brackets are essential in a test like
print "false\n" if (8 | 2) != 10;
=head2 C-style Logical And
X X X
Binary "&&" performs a short-circuit logical AND operation. That is,
if the left operand is false, the right operand is not even evaluated.
Scalar or list context propagates down to the right operand if it
is evaluated.
=head2 C-style Logical Or
X X
Binary "||" performs a short-circuit logical OR operation. That is,
if the left operand is true, the right operand is not even evaluated.
Scalar or list context propagates down to the right operand if it
is evaluated.
=head2 C-style Logical Defined-Or
X/> X
Although it has no direct equivalent in C, Perl's C/> operator is related
to its C-style or. In fact, it's exactly the same as C, except that it
tests the left hand side's definedness instead of its truth. Thus, C
is similar to C (except that it returns the value of C
rather than the value of C) and yields the same result as
C (except that the ternary-operator form can be
used as a lvalue, while C cannot). This is very useful for
providing default values for variables. If you actually want to test if
at least one of C and C is defined, use C.
The C, C/> and C operators return the last value evaluated
(unlike C's C and C, which return 0 or 1). Thus, a reasonably
portable way to find out the home directory might be:
$home = $ENV{'HOME'} // $ENV{'LOGDIR'} //
(getpwuid($ and C when used for
control flow, Perl provides the C and C operators (see below).
The short-circuit behavior is identical. The precedence of "and"
and "or" is much lower, however, so that you can safely use them after a
list operator without the need for parentheses:
unlink "alpha", "beta", "gamma"
or gripe(), next LINE;
With the C-style operators that would have been written like this:
unlink("alpha", "beta", "gamma")
|| (gripe(), next LINE);
Using "or" for assignment is unlikely to do what you want; see below.
=head2 Range Operators
X X X X
Binary ".." is the range operator, which is really two different
operators depending on the context. In list context, it returns a
list of values counting (up by ones) from the left value to the right
value. If the left value is greater than the right value then it
returns the empty list. The range operator is useful for writing
C loops and for doing slice operations on arrays. In
the current implementation, no temporary array is created when the
range operator is used as the expression in C loops, but older
versions of Perl might burn a lot of memory when you write something
like this:
for (1 .. 1_000_000) {
# code
}
The range operator also works on strings, using the magical
auto-increment, see below.
In scalar context, ".." returns a boolean value. The operator is
bistable, like a flip-flop, and emulates the line-range (comma)
operator of B, B, and various editors. Each ".." operator
maintains its own boolean state, even across calls to a subroutine
that contains it. It is false as long as its left operand is false.
Once the left operand is true, the range operator stays true until the
right operand is true, I which the range operator becomes false
again. It doesn't become false till the next time the range operator
is evaluated. It can test the right operand and become false on the
same evaluation it became true (as in B), but it still returns
true once. If you don't want it to test the right operand until the
next evaluation, as in B, just use three dots ("...") instead of
two. In all other regards, "..." behaves just like ".." does.
The right operand is not evaluated while the operator is in the
"false" state, and the left operand is not evaluated while the
operator is in the "true" state. The precedence is a little lower
than || and &&. The value returned is either the empty string for
false, or a sequence number (beginning with 1) for true. The sequence
number is reset for each range encountered. The final sequence number
in a range has the string "E0" appended to it, which doesn't affect
its numeric value, but gives you something to search for if you want
to exclude the endpoint. You can exclude the beginning point by
waiting for the sequence number to be greater than 1.
If either operand of scalar ".." is a constant expression,
that operand is considered true if it is equal (C<==>) to the current
input line number (the C variable).
To be pedantic, the comparison is actually C,
but that is only an issue if you use a floating point expression; when
implicitly using C as described in the previous paragraph, the
comparison is C which is only an issue when C
is set to a floating point value and you are not reading from a file.
Furthermore, C or C<2.18 .. 3.14> will not do what
you want in scalar context because each of the operands are evaluated
using their integer representation.
Examples:
As a scalar operator:
if (101 .. 200) { print; } # print 2nd hundred lines, short for
# if ($. == 101 .. $. == 200) { print; }
next LINE if (1 .. /^$/); # skip header lines, short for
# next LINE if ($. == 1 .. /^$/);
# (typically in a loop labeled LINE)
s/^/> / if (/^$/ .. eof()); # quote body
# parse mail messages
while (<>) {
$in_header = 1 .. /^$/;
$in_body = /^$/ .. eof;
if ($in_header) {
# do something
} else { # in body
# do something else
}
} continue {
close ARGV if eof; # reset $. each file
}
Here's a simple example to illustrate the difference between
the two range operators:
@lines = (" - Foo",
"01 - Bar",
"1 - Baz",
" - Quux");
foreach (@lines) {
if (/0/ .. /1/) {
print "$_\n";
}
}
This program will print only the line containing "Bar". If
the range operator is changed to C, it will also print the
"Baz" line.
And now some examples as a list operator:
for (101 .. 200) { print; } # print $_ 100 times
@foo = @foo[0 .. $#foo]; # an expensive no-op
@foo = @foo[$#foo-4 .. $#foo]; # slice last 5 items
The range operator (in list context) makes use of the magical
auto-increment algorithm if the operands are strings. You
can say
@alphabet = ('A' .. 'Z');
to get all normal letters of the English alphabet, or
$hexdigit = (0 .. 9, 'a' .. 'f')[$num & 15];
to get a hexadecimal digit, or
@z2 = ('01' .. '31'); print $z2[$mday];
to get dates with leading zeros.
If the final value specified is not in the sequence that the magical
increment would produce, the sequence goes until the next value would
be longer than the final value specified.
If the initial value specified isn't part of a magical increment
sequence (that is, a non-empty string matching "/^[a-zA-Z]*[0-9]*\z/"),
only the initial value will be returned. So the following will only
return an alpha:
use charnames 'greek';
my @greek_small = ("\N{alpha}" .. "\N{omega}");
To get lower-case greek letters, use this instead:
my @greek_small = map { chr } ( ord("\N{alpha}") ..
ord("\N{omega}") );
Because each operand is evaluated in integer form, C<2.18 .. 3.14> will
return two elements in list context.
@list = (2.18 .. 3.14); # same as @list = (2 .. 3);
=head2 Conditional Operator
X X X X
Ternary "?:" is the conditional operator, just as in C. It works much
like an if-then-else. If the argument before the ? is true, the
argument before the : is returned, otherwise the argument after the :
is returned. For example:
printf "I have %d dog%s.\n", $n,
($n == 1) ? '' : "s";
Scalar or list context propagates downward into the 2nd
or 3rd argument, whichever is selected.
$a = $ok ? $b : $c; # get a scalar
@a = $ok ? @b : @c; # get an array
$a = $ok ? @b : @c; # oops, that's just a count!
The operator may be assigned to if both the 2nd and 3rd arguments are
legal lvalues (meaning that you can assign to them):
($a_or_b ? $a : $b) = $c;
Because this operator produces an assignable result, using assignments
without parentheses will get you in trouble. For example, this:
$a % 2 ? $a += 10 : $a += 2
Really means this:
(($a % 2) ? ($a += 10) : $a) += 2
Rather than this:
($a % 2) ? ($a += 10) : ($a += 2)
That should probably be written more simply as:
$a += ($a % 2) ? 10 : 2;
=head2 Assignment Operators
X X X<=> X X X X
X<<< <<= >>> X X X=> X X<<< >>= >>> X X/=> X
X X X
"=" is the ordinary assignment operator.
Assignment operators work as in C. That is,
$a += 2;
is equivalent to
$a = $a + 2;
although without duplicating any side effects that dereferencing the lvalue
might trigger, such as from tie(). Other assignment operators work similarly.
The following are recognized:
**= += *= &= <<= &&=
-= /= |= >>= ||=
.= %= ^= //=
x=
Although these are grouped by family, they all have the precedence
of assignment.
Unlike in C, the scalar assignment operator produces a valid lvalue.
Modifying an assignment is equivalent to doing the assignment and
then modifying the variable that was assigned to. This is useful
for modifying a copy of something, like this:
($tmp = $global) =~ tr [A-Z] [a-z];
Likewise,
($a += 2) *= 3;
is equivalent to
$a += 2;
$a *= 3;
Similarly, a list assignment in list context produces the list of
lvalues assigned to, and a list assignment in scalar context returns
the number of elements produced by the expression on the right hand
side of the assignment.
=head2 Comma Operator
X X X
Binary "," is the comma operator. In scalar context it evaluates
its left argument, throws that value away, then evaluates its right
argument and returns that value. This is just like C's comma operator.
In list context, it's just the list argument separator, and inserts
both its arguments into the list. These arguments are also evaluated
from left to right.
The C<< => >> operator is a synonym for the comma except that it causes
its left operand to be interpreted as a string if it begins with a letter
or underscore and is composed only of letters, digits and underscores.
This includes operands that might otherwise be interpreted as operators,
constants, single number v-strings or function calls. If in doubt about
this behaviour, the left operand can be quoted explicitly.
Otherwise, the C<< => >> operator behaves exactly as the comma operator
or list argument separator, according to context.
For example:
use constant FOO => "something";
my %h = ( FOO => 23 );
is equivalent to:
my %h = ("FOO", 23);
It is I:
my %h = ("something", 23);
The C<< => >> operator is helpful in documenting the correspondence
between keys and values in hashes, and other paired elements in lists.
%hash = ( $key => $value );
login( $username => $password );
=head2 Yada Yada Operator
X X X
The yada yada operator (noted C) is a placeholder for code. Perl
parses it without error, but when you try to execute a yada yada, it
throws an exception with the text C:
sub unimplemented { ... }
eval { unimplemented() };
if( $@ eq 'Unimplemented' ) {
print "I found the yada yada!\n";
}
You can only use the yada yada to stand in for a complete statement.
These examples of the yada yada work:
{ ... }
sub foo { ... }
...;
eval { ... };
sub foo {
my( $self ) = shift;
...;
}
do { my $n; ...; print 'Hurrah!' };
The yada yada cannot stand in for an expression that is part of a
larger statement since the C is also the three-dot version of the
range operator (see L). These examples of the yada
yada are still syntax errors:
print ...;
open my($fh), '>', '/dev/passwd' or ...;
if( $condition && ... ) { print "Hello\n" };
There are some cases where Perl can't immediately tell the difference
between an expression and a statement. For instance, the syntax for a
block and an anonymous hash reference constructor look the same unless
there's something in the braces that give Perl a hint. The yada yada
is a syntax error if Perl doesn't guess that the C is a
block. In that case, it doesn't think the C is the yada yada
because it's expecting an expression instead of a statement:
my @transformed = map { ... } @input; # syntax error
You can use a C inside your block to denote that the C is
a block and not a hash reference constructor. Now the yada yada works:
my @transformed = map {; ... } @input; # ; disambiguates
my @transformed = map { ...; } @input; # ; disambiguates
=head2 List Operators (Rightward)
X X
On the right side of a list operator, it has very low precedence,
such that it controls all comma-separated expressions found there.
The only operators with lower precedence are the logical operators
"and", "or", and "not", which may be used to evaluate calls to list
operators without the need for extra parentheses:
open HANDLE, "filename"
or die "Can't open: $!\n";
See also discussion of list operators in L.
=head2 Logical Not
X X
Unary "not" returns the logical negation of the expression to its right.
It's the equivalent of "!" except for the very low precedence.
=head2 Logical And
X X
Binary "and" returns the logical conjunction of the two surrounding
expressions. It's equivalent to && except for the very low
precedence. This means that it short-circuits: i.e., the right
expression is evaluated only if the left expression is true.
=head2 Logical or, Defined or, and Exclusive Or
X X
X X
X X
Binary "or" returns the logical disjunction of the two surrounding
expressions. It's equivalent to || except for the very low precedence.
This makes it useful for control flow
print FH $data or die "Can't write to FH: $!";
This means that it short-circuits: i.e., the right expression is evaluated
only if the left expression is false. Due to its precedence, you should
probably avoid using this for assignment, only for control flow.
$a = $b or $c; # bug: this is wrong
($a = $b) or $c; # really means this
$a = $b || $c; # better written this way
However, when it's a list-context assignment and you're trying to use
"||" for control flow, you probably need "or" so that the assignment
takes higher precedence.
@info = stat($file) || die; # oops, scalar sense of stat!
@info = stat($file) or die; # better, now @info gets its due
Then again, you could always use parentheses.
Binary "xor" returns the exclusive-OR of the two surrounding expressions.
It cannot short circuit, of course.
=head2 C Operators Missing From Perl
X X X
X X
Here is what C has that Perl doesn't:
=over 8
=item unary &
Address-of operator. (But see the "\" operator for taking a reference.)
=item unary *
Dereference-address operator. (Perl's prefix dereferencing
operators are typed: $, @, %, and &.)
=item (TYPE)
Type-casting operator.
=back
=head2 Quote and Quote-like Operators
X X X X X X X
X X X

X X X X X/> X X X<<< << >>>
X X
While we usually think of quotes as literal values, in Perl they
function as operators, providing various kinds of interpolating and
pattern matching capabilities. Perl provides customary quote characters
for these behaviors, but also provides a way for you to choose your
quote character for any of them. In the following table, a C represents
any pair of delimiters you choose.
Customary Generic Meaning Interpolates
'' q{} Literal no
"" qq{} Literal yes
`` qx{} Command yes*
qw{} Word list no
// m{} Pattern match yes*
qr{} Pattern yes*
s{}{} Substitution yes*
tr{}{} Transliteration no (but see below)
< module (from CPAN, and
starting from Perl 5.8 part of the standard distribution) is able
to do this properly.
There can be whitespace between the operator and the quoting
characters, except when C is being used as the quoting character.
C is parsed as the string C, while C is the
operator C followed by a comment. Its argument will be taken
from the next line. This allows you to write:
s {foo} # Replace foo
{bar} # with bar.
The following escape sequences are available in constructs that interpolate
and in transliterations.
X X X X X X X X X X X X
X
Sequence Note Description
\t tab (HT, TAB)
\n newline (NL)
\r return (CR)
\f form feed (FF)
\b backspace (BS)
\a alarm (bell) (BEL)
\e escape (ESC)
\x{263a} [1,8] hex char (example: SMILEY)
\x1b [2,8] restricted range hex char (example: ESC)
\N{name} [3] named Unicode character or character sequence
\N{U+263D} [4,8] Unicode character (example: FIRST QUARTER MOON)
\c[ [5] control char (example: chr(27))
\o{23072} [6,8] octal char (example: SMILEY)
\033 [7,8] restricted range octal char (example: ESC)
=over 4
=item [1]
The result is the character specified by the hexadecimal number between
the braces. See L[8]> below for details on which character.
Only hexadecimal digits are valid between the braces. If an invalid
character is encountered, a warning will be issued and the invalid
character and all subsequent characters (valid or invalid) within the
braces will be discarded.
If there are no valid digits between the braces, the generated character is
the NULL character (C). However, an explicit empty brace (C)
will not cause a warning.
=item [2]
The result is the character specified by the hexadecimal number in the range
0x00 to 0xFF. See L[8]> below for details on which character.
Only hexadecimal digits are valid following C. When C is followed
by fewer than two valid digits, any valid digits will be zero-padded. This
means that C will be interpreted as C and C alone will be
interpreted as C. Except at the end of a string, having fewer than
two valid digits will result in a warning. Note that while the warning
says the illegal character is ignored, it is only ignored as part of the
escape and will still be used as the subsequent character in the string.
For example:
Original Result Warns?
"\x7" "\x07" no
"\x" "\x00" no
"\x7q" "\x07q" yes
"\xq" "\x00q" yes
=item [3]
The result is the Unicode character or character sequence given by I.
See L.
=item [4]
C}> means the Unicode character whose Unicode code
point is I.
=item [5]
The character following C is mapped to some other character as shown in the
table:
Sequence Value
\c@ chr(0)
\cA chr(1)
\ca chr(1)
\cB chr(2)
\cb chr(2)
...
\cZ chr(26)
\cz chr(26)
\c[ chr(27)
\c] chr(29)
\c^ chr(30)
\c? chr(127)
Also, C> yields C< chr(28) . "I"> for any I, but cannot come at the
end of a string, because the backslash would be parsed as escaping the end
quote.
On ASCII platforms, the resulting characters from the list above are the
complete set of ASCII controls. This isn't the case on EBCDIC platforms; see
L for the complete list of what these
sequences mean on both ASCII and EBCDIC platforms.
Use of any other character following the "c" besides those listed above is
discouraged, and may become deprecated or forbidden. What happens for those
other characters currently though, is that the value is derived by inverting
the 7th bit (0x40).
To get platform independent controls, you can use C.
=item [6]
The result is the character specified by the octal number between the braces.
See L[8]> below for details on which character.
If a character that isn't an octal digit is encountered, a warning is raised,
and the value is based on the octal digits before it, discarding it and all
following characters up to the closing brace. It is a fatal error if there are
no octal digits at all.
=item [7]
The result is the character specified by the three digit octal number in the
range 000 to 777 (but best to not use above 077, see next paragraph). See
L[8]> below for details on which character.
Some contexts allow 2 or even 1 digit, but any usage without exactly
three digits, the first being a zero, may give unintended results. (For
example, see L.) Starting in Perl 5.14, you may
use C instead which avoids all these problems. Otherwise, it is best to
use this construct only for ordinals C and below, remembering to pad to
the left with zeros to make three digits. For larger ordinals, either use
C , or convert to something else, such as to hex and use C
instead.
Having fewer than 3 digits may lead to a misleading warning message that says
that what follows is ignored. For example, C in the ASCII character set
is equivalent to the two characters C, but the warning C will be thrown. To avoid this warning, make sure to pad
your octal number with C<0>s: C.
=item [8]
Several of the constructs above specify a character by a number. That number
gives the character's position in the character set encoding (indexed from 0).
This is called synonymously its ordinal, code position, or code point). Perl
works on platforms that have a native encoding currently of either ASCII/Latin1
or EBCDIC, each of which allow specification of 256 characters. In general, if
the number is 255 (0xFF, 0377) or below, Perl interprets this in the platform's
native encoding. If the number is 256 (0x100, 0400) or above, Perl interprets
it as as a Unicode code point and the result is the corresponding Unicode
character. For example C and C both are the number 80 in
decimal, which is less than 256, so the number is interpreted in the native
character set encoding. In ASCII the character in the 80th position (indexed
from 0) is the letter "P", and in EBCDIC it is the ampersand symbol "&".
C and C are both 256 in decimal, so the number is interpreted
as a Unicode code point no matter what the native encoding is. The name of the
character in the 100th position (indexed by 0) in Unicode is
C.
There are a couple of exceptions to the above rule. C}> is
always interpreted as a Unicode code point, so that C is "P" even
on EBCDIC platforms. And if L>|encoding> is in effect, the
number is considered to be in that encoding, and is translated from that into
the platform's native encoding if there is a corresponding native character;
otherwise to Unicode.
=back
B: Unlike C and other languages, Perl has no C escape sequence for
the vertical tab (VT - ASCII 11), but you may use C or C. (C
does have meaning in regular expression patterns in Perl, see L.)
The following escape sequences are available in constructs that interpolate,
but not in transliterations.
X X X X X X
\l lowercase next char
\u uppercase next char
\L lowercase till \E
\U uppercase till \E
\Q quote non-word characters till \E
\E end either case modification or quoted section
If C is in effect, the case map used by C, C,
C and C is taken from the current locale. See L.
If Unicode (for example, C or wide hex characters of 0x100 or
beyond) is being used, the case map used by C, C, C and
C is as defined by Unicode.
All systems use the virtual C to represent a line terminator,
called a "newline". There is no such thing as an unvarying, physical
newline character. It is only an illusion that the operating system,
device drivers, C libraries, and Perl all conspire to preserve. Not all
systems read C as ASCII CR and C as ASCII LF. For example,
on a Mac, these are reversed, and on systems without line terminator,
printing C may emit no actual data. In general, use C when
you mean a "newline" for your system, but use the literal ASCII when you
need an exact character. For example, most networking protocols expect
and prefer a CR+LF (C or C) for line terminators,
and although they often accept just C, they seldom tolerate just
C. If you get in the habit of using C for networking,
you may be burned some day.
X X X X
X X X
For constructs that do interpolate, variables beginning with "C"
or "C" are interpolated. Subscripted variables such as C or
C<< $href->{key}[0] >> are also interpolated, as are array and hash slices.
But method calls such as C<< $obj->meth >> are not.
Interpolating an array or slice interpolates the elements in order,
separated by the value of C, so is equivalent to interpolating
C. "Punctuation" arrays such as C are only
interpolated if the name is enclosed in braces C, but special
arrays C, C, and C are interpolated, even without braces.
For double-quoted strings, the quoting from C is applied after
interpolation and escapes are processed.
"abc\Qfoo\tbar$s\Exyz"
is equivalent to
"abc" . quotemeta("foo\tbar$s") . "xyz"
For the pattern of regex operators (C, C and C),
the quoting from C is applied after interpolation is processed,
but before escapes are processed. This allows the pattern to match
literally (except for C and C). For example, the following matches:
'\s\t' =~ /\Q\s\t/
Because C or C trigger interpolation, you'll need to use something
like C\Quser\E\@\Qhost/> to match them literally.
Patterns are subject to an additional level of interpretation as a
regular expression. This is done as a second pass, after variables are
interpolated, so that regular expressions may be incorporated into the
pattern from the variables. If this is not what you want, use C to
interpolate a variable literally.
Apart from the behavior described above, Perl does not expand
multiple levels of interpolation. In particular, contrary to the
expectations of shell programmers, back-quotes do I interpolate
within double quotes, nor do single quotes impede evaluation of
variables when used within double quotes.
=head2 Regexp Quote-Like Operators
X
Here are the quote-like operators that apply to pattern
matching and related activities.
=over 8
=item qr/STRING/msixpo
X X X X X X X
This operator quotes (and possibly compiles) its I as a regular
expression. I is interpolated the same way as I
in C. If "'" is used as the delimiter, no interpolation
is done. Returns a Perl value which may be used instead of the
corresponding C expression. The returned value is a
normalized version of the original pattern. It magically differs from
a string containing the same characters: C returns "Regexp",
even though dereferencing the result returns undef.
For example,
$rex = qr/my.STRING/is;
print $rex; # prints (?si-xm:my.STRING)
s/$rex/foo/;
is equivalent to
s/my.STRING/foo/is;
The result may be used as a subpattern in a match:
$re = qr/$pattern/;
$string =~ /foo${re}bar/; # can be interpolated in other patterns
$string =~ $re; # or used standalone
$string =~ /$re/; # or this way
Since Perl may compile the pattern at the moment of execution of qr()
operator, using qr() may have speed advantages in some situations,
notably if the result of qr() is used standalone:
sub match {
my $patterns = shift;
my @compiled = map qr/$_/i, @$patterns;
grep {
my $success = 0;
foreach my $pat (@compiled) {
$success = 1, last if /$pat/;
}
$success;
} @_;
}
Precompilation of the pattern into an internal representation at
the moment of qr() avoids a need to recompile the pattern every
time a match C$pat/> is attempted. (Perl has many other internal
optimizations, but none would be triggered in the above example if
we did not use qr() operator.)
Options (specified by the following modifiers) are:
m Treat string as multiple lines.
s Treat string as single line. (Make . match a newline)
i Do case-insensitive pattern matching.
x Use extended regular expressions.
p When matching preserve a copy of the matched string so
that ${^PREMATCH}, ${^MATCH}, ${^POSTMATCH} will be defined.
o Compile pattern only once.
If a precompiled pattern is embedded in a larger pattern then the effect
of 'msixp' will be propagated appropriately. The effect of the 'o'
modifier has is not propagated, being restricted to those patterns
explicitly using it.
See L for additional information on valid syntax for STRING, and
for a detailed look at the semantics of regular expressions.
=item m/PATTERN/msixpogc
X X
X X X X
X X X X X X X X
=item /PATTERN/msixpogc
Searches a string for a pattern match, and in scalar context returns
true if it succeeds, false if it fails. If no string is specified
via the C<=~> or C operator, the $_ string is searched. (The
string specified with C<=~> need not be an lvalue--it may be the
result of an expression evaluation, but remember the C<=~> binds
rather tightly.) See also L. See L for
discussion of additional considerations that apply when C
is in effect.
Options are as described in C; in addition, the following match
process modifiers are available:
g Match globally, i.e., find all occurrences.
c Do not reset search position on a failed match when /g is in effect.
If "/" is the delimiter then the initial C is optional. With the C
you can use any pair of non-whitespace characters
as delimiters. This is particularly useful for matching path names
that contain "/", to avoid LTS (leaning toothpick syndrome). If "?" is
the delimiter, then a match-only-once rule applies,
described in C below.
If "'" is the delimiter, no interpolation is performed on the PATTERN.
When using a character valid in an identifier, whitespace is required
after the C.
PATTERN may contain variables, which will be interpolated (and the
pattern recompiled) every time the pattern search is evaluated, except
for when the delimiter is a single quote. (Note that C, C, and
C are not interpolated because they look like end-of-string tests.)
If you want such a pattern to be compiled only once, add a C after
the trailing delimiter. This avoids expensive run-time recompilations,
and is useful when the value you are interpolating won't change over
the life of the script. However, mentioning C constitutes a promise
that you won't change the variables in the pattern. If you change them,
Perl won't even notice. See also L.
=item The empty pattern //
If the PATTERN evaluates to the empty string, the last
I matched regular expression is used instead. In this
case, only the C and C flags on the empty pattern is honoured -
the other flags are taken from the original pattern. If no match has
previously succeeded, this will (silently) act instead as a genuine
empty pattern (which will always match).
Note that it's possible to confuse Perl into thinking C/> (the empty
regex) is really C/> (the defined-or operator). Perl is usually pretty
good about this, but some pathological cases might trigger this, such as
C (is that C or C?) and C
(C or C?). In all of these examples, Perl
will assume you meant defined-or. If you meant the empty regex, just
use parentheses or spaces to disambiguate, or even prefix the empty
regex with an C (so C/> becomes C).
=item Matching in list context
If the C option is not used, C in list context returns a
list consisting of the subexpressions matched by the parentheses in the
pattern, i.e., (C, C, C...). (Note that here C etc. are
also set, and that this differs from Perl 4's behavior.) When there are
no parentheses in the pattern, the return value is the list C for
success. With or without parentheses, an empty list is returned upon
failure.
Examples:
open(TTY, '/dev/tty');
=~ /^y/i && foo(); # do foo if desired
if (/Version: *([0-9.]*)/) { $version = $1; }
next if m#^/usr/spool/uucp#;
# poor man's grep
$arg = shift;
while (<>) {
print if /$arg/o; # compile only once
}
if (($F1, $F2, $Etc) = ($foo =~ /^(\S+)\s+(\S+)\s*(.*)/))
This last example splits $foo into the first two words and the
remainder of the line, and assigns those three fields to $F1, $F2, and
$Etc. The conditional is true if any variables were assigned, i.e., if
the pattern matched.
The C modifier specifies global pattern matching--that is,
matching as many times as possible within the string. How it behaves
depends on the context. In list context, it returns a list of the
substrings matched by any capturing parentheses in the regular
expression. If there are no parentheses, it returns a list of all
the matched strings, as if there were parentheses around the whole
pattern.
In scalar context, each execution of C finds the next match,
returning true if it matches, and false if there is no further match.
The position after the last match can be read or set using the C
function; see L. A failed match normally resets the
search position to the beginning of the string, but you can avoid that
by adding the C modifier (e.g. C). Modifying the target
string also resets the search position.
=item \G assertion
You can intermix C matches with C, where C is a
zero-width assertion that matches the exact position where the
previous C, if any, left off. Without the C modifier, the
C assertion still anchors at C as it was at the start of
the operation (see L), but the match is of course only
attempted once. Using C without C on a target string that has
not previously had a C match applied to it is the same as using
the C assertion to match the beginning of the string. Note also
that, currently, C is only properly supported when anchored at the
very beginning of the pattern.
Examples:
# list context
($one,$five,$fifteen) = (`uptime` =~ /(\d+\.\d+)/g);
# scalar context
$/ = "";
while (defined($paragraph = <>)) {
while ($paragraph =~ /[a-z]['")]*[.!?]+['")]*\s/g) {
$sentences++;
}
}
print "$sentences\n";
# using m//gc with \G
$_ = "ppooqppqq";
while ($i++ < 2) {
print "1: '";
print $1 while /(o)/gc; print "', pos=", pos, "\n";
print "2: '";
print $1 if /\G(q)/gc; print "', pos=", pos, "\n";
print "3: '";
print $1 while /(p)/gc; print "', pos=", pos, "\n";
}
print "Final: '$1', pos=",pos,"\n" if /\G(.)/;
The last example should print:
1: 'oo', pos=4
2: 'q', pos=5
3: 'pp', pos=7
1: '', pos=7
2: 'q', pos=8
3: '', pos=8
Final: 'q', pos=8
Notice that the final match matched C instead of C

, which a match
without the C anchor would have done. Also note that the final match
did not update C. C is only updated on a C match. If the
final match did indeed match C

, it's a good bet that you're running an
older (pre-5.6.0) Perl.
A useful idiom for C-like scanners is C\G.../gc>. You can
combine several regexps like this to process a string part-by-part,
doing different actions depending on which regexp matched. Each
regexp tries to match where the previous one leaves off.
$_ = <new( "http://example.com/" ); die if $url eq "xXx";
EOL
LOOP:
{
print(" digits"), redo LOOP if /\G\d+\b[,.;]?\s*/gc;
print(" lowercase"), redo LOOP if /\G[a-z]+\b[,.;]?\s*/gc;
print(" UPPERCASE"), redo LOOP if /\G[A-Z]+\b[,.;]?\s*/gc;
print(" Capitalized"), redo LOOP if /\G[A-Z][a-z]+\b[,.;]?\s*/gc;
print(" MiXeD"), redo LOOP if /\G[A-Za-z]+\b[,.;]?\s*/gc;
print(" alphanumeric"), redo LOOP if /\G[A-Za-z0-9]+\b[,.;]?\s*/gc;
print(" line-noise"), redo LOOP if /\G[^A-Za-z0-9]+/gc;
print ". That's all!\n";
}
Here is the output (split into several lines):
line-noise lowercase line-noise UPPERCASE line-noise UPPERCASE
line-noise lowercase line-noise lowercase line-noise lowercase
lowercase line-noise lowercase lowercase line-noise lowercase
lowercase line-noise MiXeD line-noise. That's all!
=item m?PATTERN?
X> X
=item ?PATTERN?
This is just like the C search, except that it matches
only once between calls to the reset() operator. This is a useful
optimization when you want to see only the first occurrence of
something in each file of a set of files, for instance. Only C
patterns local to the current package are reset.
while (<>) {
if (m?^$?) {
# blank line between header and body
}
} continue {
reset if eof; # clear m?? status for next file
}
The match-once behaviour is controlled by the match delimiter being
C>; with any other delimiter this is the normal C operator.
For historical reasons, the leading C in C is optional,
but the resulting C syntax is deprecated, will warn on
usage and may be removed from a future stable release of Perl without
further notice.
=item s/PATTERN/REPLACEMENT/msixpogcer
X X X X
X X X X X X

X X X X X
Searches a string for a pattern, and if found, replaces that pattern
with the replacement text and returns the number of substitutions
made. Otherwise it returns false (specifically, the empty string).
If the C (non-destructive) option is used then it will perform the
substitution on a copy of the string and return the copy whether or not a
substitution occurred. The original string will always remain unchanged in
this case. The copy will always be a plain string, even if the input is an
object or a tied variable.
If no string is specified via the C<=~> or C operator, the C
variable is searched and modified. (The string specified with C<=~> must
be scalar variable, an array element, a hash element, or an assignment
to one of those, i.e., an lvalue.)
If the delimiter chosen is a single quote, no interpolation is
done on either the PATTERN or the REPLACEMENT. Otherwise, if the
PATTERN contains a $ that looks like a variable rather than an
end-of-string test, the variable will be interpolated into the pattern
at run-time. If you want the pattern compiled only once the first time
the variable is interpolated, use the C option. If the pattern
evaluates to the empty string, the last successfully executed regular
expression is used instead. See L for further explanation on these.
See L for discussion of additional considerations that apply
when C is in effect.
Options are as with m// with the addition of the following replacement
specific options:
e Evaluate the right side as an expression.
ee Evaluate the right side as a string then eval the result.
r Return substitution and leave the original string untouched.
Any non-whitespace delimiter may replace the slashes. Add space after
the C when using a character allowed in identifiers. If single quotes
are used, no interpretation is done on the replacement string (the C
modifier overrides this, however). Unlike Perl 4, Perl 5 treats backticks
as normal delimiters; the replacement text is not evaluated as a command.
If the PATTERN is delimited by bracketing quotes, the REPLACEMENT has
its own pair of quotes, which may or may not be bracketing quotes, e.g.,
C or C<< s/bar/ >>. A C will cause the
replacement portion to be treated as a full-fledged Perl expression
and evaluated right then and there. It is, however, syntax checked at
compile-time. A second C modifier will cause the replacement portion
to be Ced before being run as a Perl expression.
Examples:
s/\bgreen\b/mauve/g; # don't change wintergreen
$path =~ s|/usr/bin|/usr/local/bin|;
s/Login: $foo/Login: $bar/; # run-time pattern
($foo = $bar) =~ s/this/that/; # copy first, then change
($foo = "$bar") =~ s/this/that/; # convert to string, copy, then change
$foo = $bar =~ s/this/that/r; # Same as above using /r
$foo = $bar =~ s/this/that/r
=~ s/that/the other/r; # Chained substitutes using /r
@foo = map { s/this/that/r } @bar # /r is very useful in maps
$count = ($paragraph =~ s/Mister\b/Mr./g); # get change-count
$_ = 'abc123xyz';
s/\d+/$&*2/e; # yields 'abc246xyz'
s/\d+/sprintf("%5d",$&)/e; # yields 'abc 246xyz'
s/\w/$& x 2/eg; # yields 'aabbcc 224466xxyyzz'
s/%(.)/$percent{$1}/g; # change percent escapes; no /e
s/%(.)/$percent{$1} || $&/ge; # expr now, so /e
s/^=(\w+)/pod($1)/ge; # use function call
$_ = 'abc123xyz';
$a = s/abc/def/r; # $a is 'def123xyz' and
# $_ remains 'abc123xyz'.
# expand variables in $_, but dynamics only, using
# symbolic dereferencing
s/\$(\w+)/${$1}/g;
# Add one to the value of any numbers in the string
s/(\d+)/1 + $1/eg;
# This will expand any embedded scalar variable
# (including lexicals) in $_ : First $1 is interpolated
# to the variable name, and then evaluated
s/(\$\w+)/$1/eeg;
# Delete (most) C comments.
$program =~ s {
/\* # Match the opening delimiter.
.*? # Match a minimal number of characters.
\*/ # Match the closing delimiter.
} []gsx;
s/^\s*(.*?)\s*$/$1/; # trim whitespace in $_, expensively
for ($variable) { # trim whitespace in $variable, cheap
s/^\s+//;
s/\s+$//;
}
s/([^ ]*) *([^ ]*)/$2 $1/; # reverse 1st two fields
Note the use of $ instead of \ in the last example. Unlike
B, we use the \> form in only the left hand side.
Anywhere else it's $>.
Occasionally, you can't use just a C to get all the changes
to occur that you might want. Here are two common cases:
# put commas in the right places in an integer
1 while s/(\d)(\d\d\d)(?!\d)/$1,$2/g;
# expand tabs to 8-column spacing
1 while s/\t+/' ' x (length($&)*8 - length($`)%8)/e;
=back
=head2 Quote-Like Operators
X
=over 4
=item q/STRING/
X X X X
=item 'STRING'
A single-quoted, literal string. A backslash represents a backslash
unless followed by the delimiter or another backslash, in which case
the delimiter or backslash is interpolated.
$foo = q!I said, "You said, 'She said it.'"!;
$bar = q('This is it.');
$baz = '\n'; # a two-character string
=item qq/STRING/
X X X X
=item "STRING"
A double-quoted, interpolated string.
$_ .= qq
(*** The previous line contains the naughty word "$1".\n)
if /\b(tcl|java|python)\b/i; # :-)
$baz = "\n"; # a one-character string
=item qx/STRING/
X X X X
=item `STRING`
A string which is (possibly) interpolated and then executed as a
system command with C or its equivalent. Shell wildcards,
pipes, and redirections will be honored. The collected standard
output of the command is returned; standard error is unaffected. In
scalar context, it comes back as a single (potentially multi-line)
string, or undef if the command failed. In list context, returns a
list of lines (however you've defined lines with $/ or
$INPUT_RECORD_SEPARATOR), or an empty list if the command failed.
Because backticks do not affect standard error, use shell file descriptor
syntax (assuming the shell supports this) if you care to address this.
To capture a command's STDERR and STDOUT together:
$output = `cmd 2>&1`;
To capture a command's STDOUT but discard its STDERR:
$output = `cmd 2>/dev/null`;
To capture a command's STDERR but discard its STDOUT (ordering is
important here):
$output = `cmd 2>&1 1>/dev/null`;
To exchange a command's STDOUT and STDERR in order to capture the STDERR
but leave its STDOUT to come out the old STDERR:
$output = `cmd 3>&1 1>&2 2>&3 3>&-`;
To read both a command's STDOUT and its STDERR separately, it's easiest
to redirect them separately to files, and then read from those files
when the program is done:
system("program args 1>program.stdout 2>program.stderr");
The STDIN filehandle used by the command is inherited from Perl's STDIN.
For example:
open BLAM, "blam" || die "Can't open: $!";
open STDIN, " for a clean and safe example of a manual fork() and exec()
to emulate backticks safely.
On some platforms (notably DOS-like ones), the shell may not be
capable of dealing with multiline commands, so putting newlines in
the string may not get you what you want. You may be able to evaluate
multiple commands in a single line by separating them with the command
separator character, if your shell supports that (e.g. C on many Unix
shells; C on the Windows NT C shell).
Beginning with v5.6.0, Perl will attempt to flush all files opened for
output before starting the child process, but this may not be supported
on some platforms (see L). To be safe, you may need to set
C ($AUTOFLUSH in English) or call the C method of
C on any open handles.
Beware that some command shells may place restrictions on the length
of the command line. You must ensure your strings don't exceed this
limit after any necessary interpolations. See the platform-specific
release notes for more details about your particular environment.
Using this operator can lead to programs that are difficult to port,
because the shell commands called vary between systems, and may in
fact not be present at all. As one example, the C command under
the POSIX shell is very different from the C command under DOS.
That doesn't mean you should go out of your way to avoid backticks
when they're the right way to get something done. Perl was made to be
a glue language, and one of the things it glues together is commands.
Just understand what you're getting yourself into.
See L"I/O Operators"> for more discussion.
=item qw/STRING/
X X X
Evaluates to a list of the words extracted out of STRING, using embedded
whitespace as the word delimiters. It can be understood as being roughly
equivalent to:
split(' ', q/STRING/);
the differences being that it generates a real list at compile time, and
in scalar context it returns the last element in the list. So
this expression:
qw(foo bar baz)
is semantically equivalent to the list:
'foo', 'bar', 'baz'
Some frequently seen examples:
use POSIX qw( setlocale localeconv )
@EXPORT = qw( foo bar baz );
A common mistake is to try to separate the words with comma or to
put comments into a multi-line C-string. For this reason, the
C pragma and the B switch (that is, the C variable)
produces warnings if the STRING contains the "," or the "#" character.
=item tr/SEARCHLIST/REPLACEMENTLIST/cdsr
X

X X X X X
=item y/SEARCHLIST/REPLACEMENTLIST/cdsr
Transliterates all occurrences of the characters found in the search list
with the corresponding character in the replacement list. It returns
the number of characters replaced or deleted. If no string is
specified via the =~ or !~ operator, the $_ string is transliterated. (The
string specified with =~ must be a scalar variable, an array element, a
hash element, or an assignment to one of those, i.e., an lvalue.)
If the C (non-destructive) option is used then it will perform the
replacement on a copy of the string and return the copy whether or not it
was modified. The original string will always remain unchanged in
this case. The copy will always be a plain string, even if the input is an
object or a tied variable.
A character range may be specified with a hyphen, so C

does the same replacement as C

.
For B devotees, C is provided as a synonym for C

. If the
SEARCHLIST is delimited by bracketing quotes, the REPLACEMENTLIST has
its own pair of quotes, which may or may not be bracketing quotes,
e.g., C

or C

.
Note that C

does B do regular expression character classes
such as C or C. The C

operator is not equivalent to
the tr(1) utility. If you want to map strings between lower/upper
cases, see L and L, and in general consider
using the C operator if you need regular expressions.
Note also that the whole range idea is rather unportable between
character sets--and even within character sets they may cause results
you probably didn't expect. A sound principle is to use only ranges
that begin from and end at either alphabets of equal case (a-e, A-E),
or digits (0-4). Anything else is unsafe. If in doubt, spell out the
character sets in full.
Options:
c Complement the SEARCHLIST.
d Delete found but unreplaced characters.
s Squash duplicate replaced characters.
r Return the modified string and leave the original string
untouched.
If the C modifier is specified, the SEARCHLIST character set
is complemented. If the C modifier is specified, any characters
specified by SEARCHLIST not found in REPLACEMENTLIST are deleted.
(Note that this is slightly more flexible than the behavior of some
B

programs, which delete anything they find in the SEARCHLIST,
period.) If the C modifier is specified, sequences of characters
that were transliterated to the same character are squashed down
to a single instance of the character.
If the C modifier is used, the REPLACEMENTLIST is always interpreted
exactly as specified. Otherwise, if the REPLACEMENTLIST is shorter
than the SEARCHLIST, the final character is replicated till it is long
enough. If the REPLACEMENTLIST is empty, the SEARCHLIST is replicated.
This latter is useful for counting characters in a class or for
squashing character sequences in a class.
Examples:
$ARGV[1] =~ tr/A-Z/a-z/; # canonicalize to lower case
$cnt = tr/*/*/; # count the stars in $_
$cnt = $sky =~ tr/*/*/; # count the stars in $sky
$cnt = tr/0-9//; # count the digits in $_
tr/a-zA-Z//s; # bookkeeper -> bokeper
($HOST = $host) =~ tr/a-z/A-Z/;
$HOST = $host =~ tr/a-z/A-Z/r; # same thing
$HOST = $host =~ tr/a-z/A-Z/r # chained with s///
=~ s/:/ -p/r;
tr/a-zA-Z/ /cs; # change non-alphas to single space
@stripped = map tr/a-zA-Z/ /csr, @original;
# /r with map
tr [\200-\377]
[\000-\177]; # delete 8th bit
If multiple transliterations are given for a character, only the
first one is used:
tr/AAA/XYZ/
will transliterate any A to X.
Because the transliteration table is built at compile time, neither
the SEARCHLIST nor the REPLACEMENTLIST are subjected to double quote
interpolation. That means that if you want to use variables, you
must use an eval():
eval "tr/$oldlist/$newlist/";
die $@ if $@;
eval "tr/$oldlist/$newlist/, 1" or die $@;
=item < X X X<<< << >>>
A line-oriented form of quoting is based on the shell "here-document"
syntax. Following a C<< << >> you specify a string to terminate
the quoted material, and all lines following the current line down to
the terminating string are the value of the item.
The terminating string may be either an identifier (a word), or some
quoted text. An unquoted identifier works like double quotes.
There may not be a space between the C<< << >> and the identifier,
unless the identifier is explicitly quoted. (If you put a space it
will be treated as a null identifier, which is valid, and matches the
first empty line.) The terminating string must appear by itself
(unquoted and with no surrounding whitespace) on the terminating line.
If the terminating string is quoted, the type of quotes used determine
the treatment of the text.
=over 4
=item Double Quotes
Double quotes indicate that the text will be interpolated using exactly
the same rules as normal double quoted strings.
print <
being treated as two backslashes and not one as they would in every
other quoting construct.
This is the only form of quoting in perl where there is no need
to worry about escaping content, something that code generators
can and do make good use of.
=item Backticks
The content of the here doc is treated just as it would be if the
string were embedded in backticks. Thus the content is interpolated
as though it were double quoted and then executed via the shell, with
the results of the execution returned.
print << `EOC`; # execute command and get results
echo hi there
EOC
=back
It is possible to stack multiple here-docs in a row:
print <.
chomp($string = <,
the quoted material must come on the lines following the final delimiter.
So instead of
s/this/<